Dynare Tutorial

Sun-Bin Kim
Yonsei University

Lecture at Bank of Korea

Calibrated Charateristic Equations with Level Variables

$$\frac{1}{c} = \beta E\left[\frac{1}{c'}\left(1 + r'\right)\right]$$

$$\frac{A}{1-h} = \frac{w}{c}$$

$$a' = (1+r)a + wh - c$$

$$r = \alpha z\left(\frac{K}{L}\right)^{\alpha-1} - \delta$$

$$w = (1-\alpha)z\left(\frac{K}{L}\right)^{\alpha}$$

$$\log z' = \rho \log z + \epsilon', \quad \epsilon \sim N(0, \sigma_\epsilon^2)$$

$$a = K, \quad h = L$$

Preamble Block (level variables)

```
// preamble: declare variables and parameters

var prod cons invst cap labor irate wage asst hour z;
varexo eps;
parameters bta delta alpha A rho sigeps;


// set parameter values

bta = 0.99;
delta = 0.025;
alpha = 0.36;
A = 1.723;
rho = 0.95;
sigeps = 0.007;
```

Model Block (level variables)

```
// specify model equations

model;
1/cons = bta * 1/cons(+1) * (1 + irate(+1));
A*cons/(1-hour) = wage;
asst = wage*hour + (1+irate)*asst(-1) - cons;
cap = asst(-1);
labor = hour;
irate = alpha * z(-1) * (cap/labor)^(alpha-1) - delta;
wage  = (1-alpha) * z(-1) * (cap/labor)^alpha;
prod  = z(-1) * cap^alpha * labor^(1-alpha);
invst = wage*hour + (irate+delta)*asst(-1) - cons;
log(z) = rho*log(z(-1)) + eps;
end;
```

Steady State Block (level variables)

```
// set initial values of endogenous variables

initval;
prod = 1.2;
cons = 0.9;
invst = 0.3;
cap = 12;
labor = 0.33;
irate = 0.01;
wage = 2.4;
asst = 12;
hour = 0.33;
z = 1;
end;

// compute steady state

steady;
```

Shocks + Computation

```
// specify shocks

shocks;
var eps = sigeps^2;
end;


// compute solution

stoch_simul(order=1, hp_filter=1600,
            periods=40000, drop = 1000, irf = 80);
```

Dynare Results

See Matlab Outputs

Technology Shock



Impulse Responses to Technology Shock



Impulse Responses in % Dev. from Steady State



Social Planner's Problem

The benevolent social planner maximizes the consumer's life-time utility given the resource constraints:

$$
\begin{aligned}
V(K,z) &= \max_{C,L} \left\{ u(C,L) + \beta E\left[V(K',z')|z\right] \right\} \qquad (1) \\
\text{s.t.} \;\; K' &= F(z,K,L) + (1-\delta)K - c \\
z' &= g(z,\epsilon)
\end{aligned}
$$

- The social planner's allocation is identical to that from the competitive equilibrium.
- By definition, the planner's solution is Pareto efficient.
- One can set $w = F_L(K,K,z)$ and $r = F_K(F,L,z) - \delta$ to construct a competitive equilibrium from the social planner's allocation.

Calibrated Charateristic Equations (Planner)

$$\frac{1}{C} = \beta E\left[\frac{1}{C'}\left(1-\delta+\alpha z\left(\frac{K}{L}\right)^{\alpha-1}\right)\right]$$

$$\frac{AC}{1-L} = (1-\alpha)z\left(\frac{K}{L}\right)^{\alpha}$$

$$K' = zK^{\alpha}L^{1-\alpha} + (1-\delta)K - C$$

$$\log z' = \rho \log z + \epsilon', \quad \epsilon \sim N(0, \sigma_{\epsilon}^2)$$

Preamble (Planner economy)

```
// preamble: declare variables and parameters

var prod cons invst cap labor irate wage z;
varexo eps;
parameters bta delta alpha A rho sigeps;


// set parameter values

bta = 0.99;
delta = 0.025;
alpha = 0.36;
A = 1.723;
rho = 0.95;
sigeps = 0.007;
```

Model (planner economy)

```
// specify model equations

model;
1/cons = bta * 1/cons(+1) * (1 + irate(+1));
A*cons/(1-labor) = wage;
prod  = z(-1) * cap(-1)^alpha * labor^(1-alpha);
cap = prod + (1-delta)*cap(-1) - cons;
invst = cap - (1-delta)*cap(-1);
log(z) = rho*log(z(-1)) + eps;
irate = alpha * z(-1) * (cap(-1)/labor)^(alpha-1) - delta;
wage  = (1-alpha) * z(-1) * (cap(-1)/labor)^alpha;
end;
```

Dynare Results: planner economy

See Matlab Outputs

Impulse Responses (planner economy)



Imp. Resp. in %Δ from S.S. (planner economy)



Log-Linearized Characteristic Equations

$$0 = E\left[\hat{c}_t - \hat{c}_{t+1} + \frac{\bar{r}}{1+\bar{r}}\hat{r}_{t+1}\right]$$

$$\hat{h}_t = \frac{1-\bar{h}}{\bar{h}}\left(\hat{w}_t - \hat{c}_t\right)$$

$$\hat{c}_t = \phi_a\left(\hat{r}_t + \hat{a}_t\right) + (1-\phi_a)\left(\hat{w}_t + \hat{h}_t\right) + \left(\frac{1}{\bar{r}}\right)\phi_a\left(\hat{a}_t - \hat{a}_{t+1}\right)$$

$$\hat{r}_t = \left(\frac{\bar{r}+\delta}{\bar{r}}\right)\left(\hat{z}_t + (\alpha-1)\left(\hat{K}_t - \hat{L}_t\right)\right)$$

$$\hat{w}_t = \hat{z}_t + \alpha\left(\hat{K}_t - \hat{L}_t\right)$$

$$\hat{z}_{t+1} = \rho\hat{z}_t + \epsilon_{t+1}$$

$$\hat{a}_t = \hat{K}_t, \quad \hat{h}_t = \hat{L}_t$$

Preamble Block (log deviation variables)

```
// preamble: declare variables and parameters

var prod cons invst cap labor irate wage asst hour z;
varexo eps;
parameters bta delta alpha A rho sigeps phi_a;
parameters prod_bar cons_bar invst_bar cap_bar labor_bar irate_bar hour_bar;


// set parameter values

bta = 0.99;
delta = 0.025;
alpha = 0.36;
A = 1.723;
rho = 0.95;
sigeps = 0.007;
```

Preamble Block (log deviation variables)

```
//  steady state values

labor_bar = 0.33;
hour_bar = labor_bar;
irate_bar = 1/bta - 1;
k_bar = (alpha/(irate_bar+delta))^(1/(1-alpha));
cap_bar = k_bar*labor_bar;
prod_bar = k_bar^alpha*labor_bar;
invst_bar = delta*cap_bar;
cons_bar = prod_bar - invst_bar;
phi_a = irate_bar*cap_bar/cons_bar;
```

Model Block (log deviation variables)

```
// specify model equations

model;
0 = cons - cons(+1) + irate_bar/(1+irate_bar) * irate(+1);
hour = (1-hour_bar)/hour_bar * (wage - cons);
cons = phi_a*(irate + asst(-1)) + (1-phi_a)*(wage+hour)
     + (1/irate_bar)*phi_a*(asst(-1) - asst);
irate = (irate_bar+delta)/irate_bar * (z(-1)
      + (alpha-1)*(cap - labor));
wage  = z(-1) + alpha*(cap - labor);
cap = asst(-1);
labor = hour;
z = rho*z(-1) + eps;
prod  = z(-1) + alpha*cap + (1-alpha)*labor;
invst_bar*invst = prod_bar*prod - cons_bar*cons;
end;
```

Steady State Block (log deviation variables)

```
// set initial values of endogenous variables

initval;
prod = 0.0;
cons = 0.0;
invst = 0.0;
cap = 0.0;
labor = 0.0;
irate = 0.01;
wage = 0.0;
asst = 0.0;
hour = 0.0;
z = 0;
end;

// compute steady state

steady;
```

Shocks + Computation

```
// specify shocks

shocks;
var eps = sigeps^2;
end;


// compute solution

stoch_simul(hp_filter=1600, periods=40000, drop = 1000, irf = 80);
//stoch_simul(hp_filter=1600);
```

Dynare Results: Log-linearized economy

- Dynare computes and simulates
  - steady state values
  - coefficients of the system of log-linearized equations as functions of state variables and a shock.
  - theoretical or simulated moments
    - standard deviations
    - cross-correlations
    - autocorrelations
  - time-series of variables with random shocks
  - impulse responses functions

Dynare Results: Log-linearized economy

See Matlab Outputs

Impulse Responses (log-linearized system)



A Variation: CRRA Utility

- The preferfences of the worker/consumer has changed as follows:

$$\max_{\{c_t, l_t\}_{t=0}^{\infty}} E_0 \sum_{t=0}^{\infty} \beta^t \left\{ \frac{c_t^{1-\sigma} - 1}{1-\sigma} - B \frac{h_t^{1+1/\gamma} - 1}{1+1/\gamma} \right\},$$

- All other elements (production technology, productivity shock process and etc.) are the same as those in the standard DSGE model.

A Variation: CRRA Utility

- Write down the characteristic equations and log-linearize them around the steady steate.
- Write the Dynare codes to simulate the equilibrium with level variables.
- Write the Dynare codes to simulate the equilibrium with log-deviation variables.

Dynare: Deterministic Model

Dynare can solve for the deterministic model to simulate the effects of

- (Un)anticipated changes in productivity
- (Un)anticipated shifts in parameters, which can be interpreted as changes in policies.

Preamble Block (Deterministic Model)

```
// preamble: declare variables and parameters

var prod cons invst cap labor irate wage asst hour;
varexo z;
parameters bta delta alpha A;


// set parameter values

bta = 0.99;
delta = 0.025;
alpha = 0.36;
A = 1.723;
```

Model Block (Deterministic Model)

```
// specify model equations

model;
1/cons = bta * 1/cons(+1) * (1 + irate(+1));
A*cons/(1-hour) = wage;
asst = wage*hour + (1+irate)*asst(-1) - cons;
cap = asst(-1);
labor = hour;
irate = alpha * z(-1) * (cap/labor)^(alpha-1) - delta;
wage  = (1-alpha) * z(-1) * (cap/labor)^alpha;
prod  = z(-1) * cap^alpha * labor^(1-alpha);
invst = prod - cons;
end;
```

Steady State Block (Deterministic Model)

```
// set initial values of endogenous variables
// and compute steady state

initval;
prod = 1.2;
cons = 0.9;
invst = 0.3;
cap = 12;
labor = 0.33;
irate = 0.01;
wage = 2.4;
asst = 12;
hour = 0.33;
z = 1;
end;
steady;
```

Temporary Technology Shocks (Deterministic Model)

```
// specify deterministic shocks

shocks;

var z;
periods 1:8;
values 0;

end;
```

Permanent Technology Shocks (Deterministic Model)

```
// set a permanent technology shock and compute the corresponding
// steady state regarded as the terminal values of endog. variables

endval;
prod = 1.2;
cons = 0.9;
invst = 0.3;
cap = 12;
labor = 0.33;
irate = 0.01;
wage = 2.4;
asst = 12;
hour = 0.33;
z = 1.1;
end;
steady;
```

Computation (Deterministic Model)

```
// compute solution

simul(periods=2000);
```

Dynare Results for a Deterministic Model

- Computes the steady state of endogenous variables
- Simulates the transitions of endogenous variables to
  - a temporary technology shock
  - an anticipated temporary technology shock
  - a permanent technology shock
  - an anticipated permanent technology shock

An Unanticipated Temporary Technology Shock



An Anticipated Temporary Technology Shock



An Unanticipated Permanent Technology Shock



An Anticipated Permanent Technology Shock